Maple-Worksheet for computing Edgeworth polynomials for for Student's t-statistic according to Hall's (1992) smooth function approach.

Authors: Helmut Finner and Thorsten Dickhaus.

Last updated: February 23, 2009

Students t-statistic defined by the one-Sample t-statistic
with nu=n-1 degrees of freedom.

Edgeworth expansion has the form:

F_n (t) = Phi(t) + \sum_{i=1}^r p_i (t) phi (t)

with

Phi: cdf of standard normal

phi: pdf of standard normal

p_i (t) depends on moments alpha[3], ... , alpha[i+2]
of the underlying universe

References:

Finner, H., Dickhaus, T. (2009).
EDGEWORTH EXPANSIONS AND RATES OF
CONVERGENCE FOR NORMALIZED SUMS:
CHUNG'S 1946 METHOD REVISITED.
Preprint.

Bartlett, M. S. (1935). The effect of non-normality on the t distribution. Proc. Camb. Philos. Soc. 31, 223-231.
Chung, K.-L. (1946). The approximate distribution of Student's statistics. Ann. Math. Stat. 17, 447-465.
Gayen, A. K. (1949). The distribution of 'Student's' t in random samples of any size drawn from non-normal universes. Biometrika 36, 353-369.
Hall, P. (1987). Edgeworth expansion for Student's t statistic under minimal moment conditions. Ann. Prob. 15, 920-931.
Hall, P. (1992). The bootstrap and Edgeworth expansion. Springer Series in Statistics,

New York.
Hsu, P. L. (1945). The approximate distributions of the mean and variance of a sample of independent variables. Ann. Math. Stat. 16, 1-29.
Wallace, D. L. (1958). Asymptotic approximations to distributions. Ann. Math. Stat. 29, 635-654.

```
> restart:
> Digits:=30:
  with(combinat):
```

USER INPUT: number of approximation polynomials needed

(Number_polynomials >= 6 may fail due to computer resources.)

```
> number_polynomials:=5;
```

$$number\_polynomials := 5 \tag{1}$$

Initialize data structures

```
> mu1:=[alpha[i],i=1..k]:
  alpha[1]:=0:
  alpha[2]:=1:
  for i from number_polynomials+3 to 1000 do:
    alpha[i]:=0:
  end:
  EEE[0,0]:=1:
  ks_sum:= 2*(number_polynomials+1);
```

$$ks\_sum := 12 \tag{2}$$

Define shortcuts for combinatorial functions

```
> z:=(kkk,i2)->partition(kkk)[i2]:
  z1:=(i1,nn,mm,kk)->(binomial(nn,mm)*(multinomial(kk,op(partition
  (kk)[i1])))*numbperm((partition(kk)[i1]))):
```

Pre-compute expectations of statistics of the form

(\sum_{i=1}^n X_i)^r * (\sum_{i=1}^n X_j^2)^s

in terms of \alpha[i] and n

This may take a long time for k >= 6 and may be impossible because of computer memory.

It should work well for k <= 4. For k=5 it may take up to one hour.

```
> for k from 1 to ks_sum do;
    #Level 0
    print(k, 0);
    EE:=0;
    test:=0;
    for i from 1 to numbpart(k) do:
       m:=nops(partition(k)[i]):
       test:=test+z1(i,n,m,k);
       ee:=1:
       for j from 1 to m do:
         j1:=partition(k)[i][j];
         ee:=ee*alpha[j1];
       end:
       ee:=ee*z1(i,n,m,k);
       EE:=EE+ee;
    end:
    EEE[k,0]:=simplify(expand(EE));


    #Level 1
    print(k, 1);
    it0:=numbpart(k):
    i:='i':
    it1:=0:
    for i from 1 to it0 do:
       partmax:=max(seq(z(k,i)[u1],u1=1..nops(z(k,i))) );
       m:=nops(z(k,i));
       for j1 from 1 to partmax do;
         ki1:=0;
         for j2 from 1 to m do:
           if(z(k,i)[j2]=j1) then ki1:=ki1+1; fi;
         end:
         if(ki1>0) then
           for j2 from 1 to m do:
             if(z(k,i)[j2]=j1) then
               it1:=it1+1;
               z_neu[1,it1]:=z(k,i);
               z_neu[1,it1][j2]:=z(k,i)[j2]+2;
               z1_neu[1,it1]:=ki1*z1(i,n,m,k);
               j2:=m+1;
```

```
          fi;
        end;
      fi;
    end;
    it1:=it1+1;
    ddd[1,it1]:=[(seq(z(k,i)[u1],u1=1..nops(z(k,i))),2)];
    ddd[1,it1];
    z_neu[1,it1]:=sort(ddd[1,it1]);
    z1_neu[1,it1]:=(n-nops(z(k,i)))*z1(i,n,m,k);
end:
test:=0:
EE:=0:
i:='i':
for i from 1 to it1 do:
  j:='j';
  m:=nops(z_neu[1,i]);
  test:=test+z1_neu[1,i];
  ee:=1:
  for j from 1 to m do:
    j1:=z_neu[1,i][j];
    ee:=ee*alpha[j1];
  end:
  ee:=ee*z1_neu[1,i];
  EE:=EE+ee;
end:
EEE[k,1]:=expand(EE);

#Levels (s+1) for s >= 1
it0:=it1:
s:='s';
i:='i';
for s from 1 to ks_sum-k do:
  print(k, s+1);
  it1:=0;
  for i from 1 to it0 do;
    partmax:=max(seq(z_neu[s,i][u1],u1=1..nops(z_neu[s,i])) );
    m:=nops(z_neu[s,i]);
    for j1 from 1 to partmax do;
      ki1:=0;
      for j2 from 1 to m do:
        if(z_neu[s,i][j2]=j1) then ki1:=ki1+1; fi;
      end:
```

```
        if(ki1>0) then
           for j2 from 1 to m do:
              if(z_neu[s,i][j2]=j1) then
                 it1:=it1+1;
                 z_neu[s+1,it1]:=z_neu[s,i];
                 z_neu[s+1,it1][j2]:=z_neu[s,i][j2]+2;
                 z1_neu[s+1,it1]:=ki1*z1_neu[s,i];
                 j2:=m+1;
              fi;
           end;
        fi;
      end;
      it1:=it1+1;
      ddd[s+1,it1]:=[(seq(z_neu[s,i][u1],u1=1..nops(z_neu[s,i])
),2)];
      ddd[s+1,it1];
      z_neu[s+1,it1]:=sort(ddd[s+1,it1]);
      z1_neu[s+1,it1]:=(n-nops(z_neu[s,i]))*z1_neu[s,i];
    end;
    test:=0;
    EE:=0;
    i:='i';
    for i from 1 to it1 do:
      j:='j';
      m:=nops(z_neu[s+1,i]);
      test:=test+z1_neu[s+1,i];
      ee:=1:
      for j from 1 to m do:
        j1:=z_neu[s+1,i][j];
        ee:=ee*alpha[j1];
      end:
      ee:=ee*z1_neu[s+1,i];
      EE:=EE+ee;
    end:
    EEE[k,s+1]:=expand(EE);
    it0:=it1;
  end:
end:
```

$$1, 0 \qquad (3)$$
$$1, 1$$
$$1, 2$$
$$1, 3$$
$$1, 4$$

1, 5
1, 6
1, 7
1, 8
1, 9
1, 10
1, 11
1, 12
2, 0
2, 1
2, 2
2, 3
2, 4
2, 5
2, 6
2, 7
2, 8
2, 9
2, 10
2, 11
3, 0
3, 1
3, 2
3, 3
3, 4
3, 5
3, 6
3, 7
3, 8
3, 9
3, 10
4, 0
4, 1
4, 2
4, 3
4, 4
4, 5
4, 6
4, 7
4, 8
4, 9
5, 0
5, 1
5, 2
5, 3
5, 4
5, 5
5, 6
5, 7
5, 8
6, 0
6, 1

6, 2
6, 3
6, 4
6, 5
6, 6
6, 7
7, 0
7, 1
7, 2
7, 3
7, 4
7, 5
7, 6
8, 0
8, 1
8, 2
8, 3
8, 4
8, 5
9, 0
9, 1
9, 2
9, 3
9, 4
10, 0
10, 1
10, 2
10, 3
11, 0
11, 1
11, 2
12, 0
12, 1

Combine the moment results across the different levels

```
> EVW:=(kkk,sss)->subs(n=1/tau^2,expand(n^(-sss-kkk)*sum(binomial
  (sss,j3)*(-n)^(sss-j3)*EEE[kkk,j3],j3=0..sss)));
```

$$EVW := (kkk, sss) \to subs\left(n = \frac{1}{\tau^2}, expand\left(n^{-sss-kkk}\left(\sum_{j3=0}^{sss} combinat\text{:-binomial}(sss, j3)\ (-n)^{sss-j3}\ EEE_{kkk,\ j3}\right)\right)\right) \quad \textbf{(4)}$$

Define the smooth function model

```
> Z:='Z';
  m:='m';
```

```
A:=(x1,x2,m)->((x1-m)/sqrt(x2-x1^2));
```

$$Z := Z$$
$$m := m$$
$$A := (x1, x2, m) \rightarrow \frac{x1 - m}{\sqrt{x2 - x1^2}} \quad \textbf{(5)}$$

Compute S[n, 1],...,S[n, j] according to Hall's approach, j= number_polynomials+1

```
> with(combinat):

  for nnnn from 1 to number_polynomials+1 do:
  i:='i':
  j:='j':
  i1:='i1':

  One1:=[seq(1,j=1..nnnn)]:
  test:=[One1]:

  for i1 from 1 to nnnn do:
    One1[i1]:=2:
    test:=[op(test),op(permute(One1))]:
  end:

  S_sum:=0:
  i3:='i3':
  i3:='i3':
  for i2 from 1 to 2^nnnn do:
    deriv:=simplify(D[op(test[i2])](A)):
    a[op(test[i2])]:=deriv(0,1,0):
    S_sum:=S_sum+tau^(nnnn-1)*1/nnnn!*a[op(test[i2])]*product(Z[
  [op(test[i2])][i3]],i3=1..nnnn):
  end:
  S[n,nnnn]:=S_sum:
  print("S[n,",nnnn,"]=",S[n,nnnn]);
  end:
```

" **(6)**

S[n,", 1, "]=", Z_1

"

S[n,", 2, "]=", $-\frac{1}{2}\tau Z_2 Z_1$

"

S[n,", 3, "]=", $\frac{1}{2}\tau^2 Z_1^3 + \frac{3}{8}\tau^2 Z_2^2 Z_1$

"

S[n,", 4, "]=", $-\frac{3}{4}\tau^3 Z_2 Z_1^3 - \frac{5}{16}\tau^3 Z_2^3 Z_1$

"

S[n,", 5, "]=", $\frac{3}{8}\tau^4 Z_1^5 + \frac{15}{16}\tau^4 Z_2^2 Z_1^3 + \frac{35}{128}\tau^4 Z_2^4 Z_1$

"

S[n,", 6, "]=", $-\frac{15}{16}\tau^5 Z_2 Z_1^5 - \frac{35}{32}\tau^5 Z_2^3 Z_1^3 - \frac{63}{256}\tau^5 Z_2^5 Z_1$

Compute S_n as sum over S[n, k], according to Hall's approach

```
> S[n]:= add(S[n,runn], runn=1..(number_polynomials+1)):
  g:=subs(Z[1]=V1/tau,Z[2]=V2/tau,S[n]):
  Sn:=unapply(g,[V1,V2]):
```

Expand powers of S_n, save results in beta[j]

```
> mm1:='mm1': iii:='iii': iiii:='iiii':
  jjj:='jjj': rr1:='rr1': rr2:='rr2':
  beta1:='beta1': beta11:='beta11':

  for mm1 from 1 to number_polynomials+2 do:
    rr1:=convert(expand((Sn(V1,V2))^mm1),polynom,ascending);
    for iii from degree(rr1, V1) to 1 by -1 do:
      for jjj from degree(rr1, V2) to 1 by -1 do:
        if (iii+jjj>ks_sum+1) then
          rr2:=algsubs(V1^iii*V2^jjj=0,rr1);
          rr1:=rr2;
        fi;
        if (iii+jjj<ks_sum+2) then
          rr2:=algsubs(V1^iii*V2^jjj=EVW(iii,jjj),rr1);
          rr1:=rr2;
        fi;
      end:
    end:
    for iiii from degree(rr1, V1) to ks_sum+2 by -1 do:
      rr2:=algsubs(V1^iiii=0,rr1);
      rr1:=rr2;
    end:
    for iiii from ks_sum+1 to 1 by -1 do:
```

```
      rr2:=algsubs(V1^iiii=EVW(iiii,0),rr1);
      rr1:=rr2;
    end:
    rr1:=collect((expand(simplify(expand(rr2)))),tau);
    beta11[mm1]:=rr1;
    beta1[mm1]:=unapply(convert(series(rr1,tau=0,
  number_polynomials+2),polynom),tau);
    beta[mm1]:=convert(beta1[mm1](tau),polynom,tau);
  end:
```

Compute kappa[k] as linear combinations of the beta[k] according to Hall's approach

```
> m:='m':
  n:='n':
  j:='j':
  t:='t':
  i_ende:=number_polynomials+2:
  kappa:='kappa':
  f1_moment:=(t,n)->log(1+sum(beta[j]/j!*(I*t)^j,j=1..i_ende)):
  f1_moment(t,n):
  equation2:=unapply(convert(series(f1_moment(t,n),t=0,i_ende+1),
  polynom,t),t):
  for i from 1 to i_ende do:
    kap11[i]:=expand(coeff(equation2(t),t^i)*i!/I^i);
  end:
> i:='i':
  for i from 1 to number_polynomials+2 do:
    kappa_1[i]:=series(kap11[i],tau=0, number_polynomials+2):
  end:
```

Expand necessary kappa's

```
> for i12 from 1 to number_polynomials+2 do:
    vv0:=simplify(kappa_1[i12]);
    vv1:=expand(tau^(-(i12-2))*vv0):
    vv2:=series(vv1,tau=0,number_polynomials+2);
    vv3:=convert(expand(series(vv1,tau=0, number_polynomials+2)),
```

```
polynom,t,ascending);
  vv4:=sort(collect(algsubs(tau^2=tau1,vv3),tau1),plex);
  kappa_1[i12]:=vv4;
end:
```

Define modified Hermite polynomials

```
> with(orthopoly):
x:='x':
H1:=(n,x)->simplify((2^(-n/2)*HermiteH(n,x/sqrt(2)))):
```

Approximation of self-normalized statistic via characteristic function

```
> k_ji:='k_ji':
i:='i':
j:='j':
assume(t::real):
kap[j,n]:=n^(-(j-2)/2)*sum(k_ji[j,i]/n^(i-1),i=1..
number_polynomials+2);
k_ji[1,1]:=0:
k_ji[2,1]:=1:
tt:=collect(sum(kap[j,n]*It[j]/j!,j=1..ks_sum)-It[2]/2,n):
tt1:=(exp(tt)):
tt2:=series(tt1,n=infinity, number_polynomials+2):
tt3:=convert(tt2-1,polynom):
hh1:=unapply(tt3,n):
assume(lambda>0):
hh2:=lambda->convert(simplify(hh1(1/lambda^2)),polynom,lambda):
hh2(lambda):
hh2a:=hh2(lambda):
```

$$kap_{j,n} := n^{-\frac{j}{2}+1}\left(k\_ji_{j,1} + \frac{k\_ji_{j,2}}{n} + \frac{k\_ji_{j,3}}{n^2} + \frac{k\_ji_{j,4}}{n^3} + \frac{k\_ji_{j,5}}{n^4} + \frac{k\_ji_{j,6}}{n^5} + \frac{k\_ji_{j,7}}{n^6}\right) \quad \textbf{(7)}$$

Finally: Compute approximation polynomials

```
> for i from 1 to number_polynomials do:
  hh3:=simplify(hh2a/lambda):
  hh4:=unapply(hh3,lambda):
  r[i]:=hh4(0):
  hh2a:=simplify((hh2a-r[i]*lambda)/lambda):
end:
for i from 1 to number_polynomials do:
  tt3:=expand(r[i]):
```

```
  for j from 20 to 1 by -1 do:
     tt4:=subs(It[j]=(I*t)^j,tt3):
     tt3:=tt4:
  end:
  p[i]:=tt3;
  p_sort[i]:=collect(convert(expand(p[i]),polynom,t),t);
  for j1 from degree(tt3, t) to 1 by -1 do;
     tt4:=subs(t^j1=-H1(j1-1,x)/I^j1,tt3);
     tt3:=tt4:
  end;
  p[i]:=tt3;
end:
```

```
> for mm2 from 1 to number_polynomials+2 do:
    rr3a:=kappa_1[mm2]:
    for mm3 from 1 to number_polynomials+2 do;
      k_ji[mm2,mm3]:=coeff(rr3a,tau1,mm3-1);
    end;
  end:
```

```
> for it1 from 1 to number_polynomials do;
    p[it1]:=subs(x=y,expand(p[it1]));
  end:
```

Display polynomials in factorized form

```
> i:='i':
  j1:='j1':
  for i from 1 to number_polynomials do:
   tt1:=seq(alpha[j1],j1=3..i+2);
   tt2:=collect(expand(p[i]),tt1);
   #print();
   print("Polynomial number", i);
   #print();
   print(p[i]);
  end:
```

"Polynomial number", 1 **(8)**

$$\frac{1}{6}\alpha_3+\frac{1}{3}\alpha_3 y^2$$

"Polynomial number", 2

$$\frac{1}{12}\alpha_4 y^3-\frac{1}{4}y\,\alpha_4-\frac{y^3}{2}-\frac{1}{9}\alpha_3^2 y^3+\frac{1}{6}\alpha_3^2 y-\frac{1}{18}\alpha_3^2 y^5$$

"Polynomial number", 3

$$-\frac{1}{16}\alpha_3-\frac{1}{40}\alpha_5+\frac{5}{48}\alpha_4\alpha_3+\frac{5}{24}\alpha_4\alpha_3 y^4+\frac{5}{8}\alpha_4\alpha_3 y^2-\frac{1}{36}\alpha_3 y^6\alpha_4-\frac{35}{432}\alpha_3^3$$

$$+\frac{1}{162}\alpha_3^3 y^8+\frac{7}{324}y^6\alpha_3^3+\frac{1}{6}\alpha_3 y^6-\frac{35}{216}\alpha_3^3 y^4-\frac{1}{20}\alpha_5 y^4-\frac{1}{4}\alpha_3 y^4-\frac{1}{5}\alpha_5 y^2-\frac{175}{432}\alpha_3^3 y^2-\frac{1}{8}\alpha_3 y^2$$

"Polynomial number", 4

$$-\frac{5}{12}\alpha_3^2 y^5\alpha_4+\frac{1}{6}\alpha_3^2 y^3\alpha_4+\frac{2}{15}\alpha_3 y^5\alpha_5-\frac{1}{18}\alpha_3^2\alpha_4 y^7-\frac{1}{12}\alpha_3 y^3\alpha_5+\frac{29}{24}\alpha_3^2 y\alpha_4-\frac{1}{2}\alpha_3 y\alpha_5+\frac{1}{216}\alpha_3^2 y^9\alpha_4+\frac{1}{60}\alpha_3 y^7\alpha_5+\frac{1}{24}\alpha_4 y^3-\frac{1}{6}\alpha_3^2 y+\frac{1}{4}y\alpha_4+\frac{11}{36}\alpha_3^2 y^5-\frac{5}{216}\alpha_3^4 y^3+\frac{25}{108}\alpha_3^4 y^5-\frac{35}{72}\alpha_3^4 y-\frac{1}{4}y^5\alpha_4-\frac{1}{45}\alpha_6 y^5+\frac{1}{18}\alpha_6 y^3+\frac{1}{6}y\alpha_6-\frac{1}{288}\alpha_4^2 y^7+\frac{7}{96}\alpha_4^2 y^5-\frac{37}{96}y\alpha_4^2-\frac{11}{96}\alpha_4^2 y^3+\frac{1}{24}y^7\alpha_4+\frac{1}{12}y^7\alpha_3^2+\frac{5}{108}\alpha_3^4 y^7-\frac{5}{1944}\alpha_3^4 y^9-\frac{1}{1944}\alpha_3^4 y^{11}-\frac{1}{36}\alpha_3^2 y^9+\frac{1}{36}\alpha_3^2 y^3+\frac{3y^5}{8}-\frac{y^7}{8}$$

"Polynomial number", 5

$$\frac{25}{768}\alpha_3+\frac{3}{64}\alpha_5+\frac{1}{336}\alpha_7+\frac{7}{64}\alpha_3^2\alpha_5-\frac{385}{1152}\alpha_4\alpha_3^3-\frac{25}{128}\alpha_4\alpha_3-\frac{7}{288}\alpha_6\alpha_3-\frac{7}{192}\alpha_4\alpha_5+\frac{35}{256}\alpha_4^2\alpha_3-\frac{75}{64}\alpha_4\alpha_3 y^2-\frac{185}{192}\alpha_4\alpha_3 y^4-\frac{5}{96}\alpha_3 y^6\alpha_4-\frac{385}{1728}\alpha_3^3\alpha_4 y^6+\frac{77}{64}\alpha_5\alpha_3^2 y^2+\frac{7}{96}\alpha_3\alpha_4^2 y^6+\frac{3}{16}\alpha_3 y^8\alpha_4-\frac{385}{128}\alpha_3^3\alpha_4 y^2+\frac{11}{96}y^8\alpha_3^3\alpha_4+\frac{1}{864}\alpha_3 y^{10}\alpha_4^2+\frac{245}{192}\alpha_3\alpha_4^2 y^4+\frac{49}{480}y^6\alpha_5\alpha_3^2-\frac{385}{144}\alpha_3^3\alpha_4 y^4+\frac{11}{1296}y^{10}\alpha_3^3\alpha_4-\frac{49}{144}\alpha_3\alpha_6 y^2+\frac{35}{32}\alpha_5\alpha_3^2 y^4-\frac{1}{30}y^8\alpha_5\alpha_3^2-\frac{1}{72}\alpha_3 y^{10}\alpha_4+\frac{175}{128}\alpha_3\alpha_4^2 y^2+\frac{1}{135}\alpha_3 y^8\alpha_6-\frac{7}{192}\alpha_3 y^8\alpha_4^2-\frac{7}{216}\alpha_3 y^6\alpha_6-\frac{7}{360}y^6\alpha_4\alpha_5-\frac{49}{144}\alpha_3\alpha_6 y^4+\frac{1}{240}y^8\alpha_4\alpha_5-\frac{7}{16}y^4\alpha_4\alpha_5-\frac{1}{1944}\alpha_3^3 y^{12}\alpha_4-\frac{7}{16}\alpha_4 y^2\alpha_5-\frac{1}{360}\alpha_3^2 y^{10}\alpha_5+\frac{175}{1152}\alpha_3^3+\frac{1001}{6912}\alpha_3^5-\frac{143}{2592}\alpha_3^5 y^8+\frac{1001}{10368}\alpha_3^5 y^6+\frac{7007}{6912}\alpha_3^5 y^4+\frac{1001}{864}\alpha_3^5 y^2-\frac{1}{72}y^{10}\alpha_3^3-\frac{5}{16}\alpha_3 y^8-\frac{143}{19440}y^{10}\alpha_3^5+\frac{1}{24}\alpha_3 y^{10}-\frac{1}{40}y^8\alpha_5+\frac{1}{29160}\alpha_3^5 y^{14}+\frac{3}{56}\alpha_7 y^2+\frac{1}{252}\alpha_7 y^6+\frac{11}{168}\alpha_7 y^4+\frac{1}{324}\alpha_3^3 y^{12}+\frac{13}{58320}\alpha_3^5 y^{12}+\frac{875}{1152}\alpha_3^3 y^2+\frac{3}{8}\alpha_5 y^2+\frac{9}{32}\alpha_5 y^4+\frac{3}{32}\alpha_3 y^4+\frac{3}{16}\alpha_3 y^6+\frac{385}{576}\alpha_3^3 y^4+\frac{35}{432}y^6\alpha_3^3-\frac{17}{108}\alpha_3^3 y^8+\frac{25}{384}\alpha_3 y^2$$

Comparison with Chung's method

Chung[1],...,Chung[6] obtained with Cung's method

```
> Chung[1] := ((1/3)*y^2+1/6)*alpha[3];
  diff1 := expand(Chung[1]-p[1]);
```

$$Chung_1 := \left(\frac{1}{6} + \frac{y^2}{3}\right) \alpha_3 \quad (9)$$

$$diff1 := 0$$

```
> if number_polynomials >=2 then
    Chung[2] := (-(1/9)*y^3-(1/18)*y^5+(1/6)*y)*alpha[3]^2+(-(1/4)
  *y+(1/12)*y^3)*alpha[4]-(1/2)*y^3;
    diff2 := expand(Chung[2]-p[2]);
  end;
```

$$Chung_2 := \left(-\frac{1}{9} y^3 + \frac{1}{6} y - \frac{1}{18} y^5\right) \alpha_3^2 + \left(-\frac{1}{4} y + \frac{1}{12} y^3\right) \alpha_4 - \frac{y^3}{2} \quad (10)$$

$$diff2 := 0$$

```
> if number_polynomials >= 3 then
    Chung[3] :=   ((1/162)*y^8-35/432+(7/324)*y^6-(35/216)*y^4-
  (175/432)*y^2)*alpha[3]^3+((5/48+(5/24)*y^4+(5/8)*y^2-(1/36)*
  y^6)*alpha[4]-1/16-(1/8)*y^2+(1/6)*y^6-(1/4)*y^4)*alpha[3]+
  (-1/40-(1/20)*y^4-(1/5)*y^2)*alpha[5];
    diff3 := expand(Chung[3]-p[3]);
  end;
```

$$Chung_3 := \left(-\frac{35}{432} - \frac{35}{216} y^4 + \frac{7}{324} y^6 - \frac{175}{432} y^2 + \frac{1}{162} y^8\right) \alpha_3^3 + \left(\left(\frac{5}{48} + \frac{5}{24} y^4 + \frac{5}{8} y^2 - \frac{1}{36} y^6\right) \alpha_4 - \frac{1}{16} - \frac{y^2}{8} + \frac{y^6}{6} - \frac{y^4}{4}\right) \alpha_3 + \left(-\frac{1}{40} - \frac{1}{20} y^4 - \frac{1}{5} y^2\right) \alpha_5 \quad (11)$$

$$diff3 := 0$$

```
> if number_polynomials >= 4 then
    Chung[4] := (-(5/1944)*y^9-(5/216)*y^3-(35/72)*y+(25/108)*y^5+
  (5/108)*y^7-(1/1944)*y^11)*alpha[3]^4+((-(5/12)*y^5+(29/24)*y+
  (1/216)*y^9-(1/18)*y^7+(1/6)*y^3)*alpha[4]+(11/36)*y^5-(1/36)*
  y^9+(1/12)*y^7-(1/6)*y+(1/36)*y^3)*alpha[3]^2+((1/60)*y^7-(1/2)*
  y+(2/15)*y^5-(1/12)*y^3)*alpha[5]*alpha[3]+((7/96)*y^5-(11/96)*
  y^3-(37/96)*y-(1/288)*y^7)*alpha[4]^2+((1/24)*y^7-(1/4)*y^5+
  (1/4)*y+(1/24)*y^3)*alpha[4]+((1/18)*y^3+(1/6)*y-(1/45)*y^5)*
  alpha[6]-(1/8)*y^7+(3/8)*y^5;
    diff4 := expand(Chung[4]-p[4]);
  end;
```

$$Chung_4 := \left(\frac{25}{108} y^5 - \frac{35}{72} y - \frac{5}{216} y^3 + \frac{5}{108} y^7 - \frac{5}{1944} y^9 - \frac{1}{1944} y^{11}\right) \alpha_3^4 + \left(\left(-\frac{5}{12} y^5 + \frac{29}{24} y + \frac{1}{216} y^9 - \frac{1}{18} y^7 + \frac{1}{6} y^3\right) \alpha_4 + \frac{11 y^5}{36} - \frac{y^9}{36} + \frac{y^7}{12} - \frac{y}{6} + \frac{y^3}{36}\right) \alpha_3^2 + \left(\frac{1}{60} y^7 - \frac{1}{2} y + \frac{2}{15} y^5 - \frac{1}{12} y^3\right) \alpha_5 \alpha_3 + \left(\frac{7}{96} y^5 - \frac{11}{96} y^3 - \frac{37}{96} y - \frac{1}{288} y^7\right) \alpha_4^2 \quad (12)$$

$$+\left(\frac{1}{24}y^7-\frac{1}{4}y^5+\frac{1}{4}y+\frac{1}{24}y^3\right)\alpha_4+\left(\frac{1}{18}y^3+\frac{1}{6}y-\frac{1}{45}y^5\right)\alpha_6-\frac{y^7}{8}+\frac{3y^5}{8}$$

$$diff4 := 0$$

```
> if number_polynomials >= 5 then
    Chung[5] := ((7007/6912)*y^4+(1001/864)*y^2+1001/6912-
  (143/19440)*y^10+(13/58320)*y^12+(1/29160)*y^14+(1001/10368)*y^6
  -(143/2592)*y^8)*alpha[3]^5+((-(385/128)*y^2+(11/96)*y^8-
  (1/1944)*y^12-(385/144)*y^4+(11/1296)*y^10-385/1152-(385/1728)*
  y^6)*alpha[4]+(875/1152)*y^2+(35/432)*y^6+175/1152-(17/108)*y^8+
  (1/324)*y^12+(385/576)*y^4-(1/72)*y^10)*alpha[3]^3+((35/32)*
  y^4+7/64-(1/360)*y^10+(77/64)*y^2+(49/480)*y^6-(1/30)*y^8)*alpha
  [5]*alpha[3]^2+(((7/96)*y^6+(1/864)*y^10+35/256-(7/192)*y^8+
  (175/128)*y^2+(245/192)*y^4)*alpha[4]^2+(-(5/96)*y^6+(3/16)*y^8-
  (185/192)*y^4-(75/64)*y^2-(1/72)*y^10-25/128)*alpha[4]+(-(7/216)
  *y^6-(49/144)*y^2+(1/135)*y^8-7/288-(49/144)*y^4)*alpha[6]+
  (1/24)*y^10+(25/384)*y^2+(3/16)*y^6+25/768-(5/16)*y^8+(3/32)*
  y^4)*alpha[3]+(-(7/360)*y^6-(7/16)*y^2-7/192-(7/16)*y^4+(1/240)*
  y^8)*alpha[5]*alpha[4]+(-(1/40)*y^8+(3/8)*y^2+(9/32)*y^4+3/64)*
  alpha[5]+((3/56)*y^2+(11/168)*y^4+(1/252)*y^6+1/336)*alpha[7];
    diff5 := expand(Chung[5]-p[5]);
  end;
```

$$\begin{aligned}Chung_5 := &\left(\frac{1001}{6912}+\frac{1001}{10368}y^6+\frac{13}{58320}y^{12}+\frac{7007}{6912}y^4+\frac{1001}{864}y^2-\frac{143}{2592}y^8\right.\\&\left.+\frac{1}{29160}y^{14}-\frac{143}{19440}y^{10}\right)\alpha_3^5+\left(\left(\left(-\frac{385}{128}y^2+\frac{11}{96}y^8-\frac{1}{1944}y^{12}-\frac{385}{144}y^4\right.\right.\right.\\&\left.\left.+\frac{11}{1296}y^{10}-\frac{385}{1152}-\frac{385}{1728}y^6\right)\alpha_4+\frac{875y^2}{1152}+\frac{35y^6}{432}+\frac{175}{1152}-\frac{17y^8}{108}+\frac{y^{12}}{324}\\&\left.+\frac{385y^4}{576}-\frac{y^{10}}{72}\right)\alpha_3^3+\left(\frac{35}{32}y^4+\frac{7}{64}-\frac{1}{360}y^{10}+\frac{77}{64}y^2+\frac{49}{480}y^6-\frac{1}{30}y^8\right)\alpha_5\alpha_3^2\\&+\left(\left(\frac{7}{96}y^6+\frac{1}{864}y^{10}+\frac{35}{256}-\frac{7}{192}y^8+\frac{175}{128}y^2+\frac{245}{192}y^4\right)\alpha_4^2+\left(-\frac{5}{96}y^6\right.\right.\\&\left.+\frac{3}{16}y^8-\frac{185}{192}y^4-\frac{75}{64}y^2-\frac{1}{72}y^{10}-\frac{25}{128}\right)\alpha_4+\left(-\frac{7}{216}y^6-\frac{49}{144}y^2+\frac{1}{135}y^8\right.\\&\left.\left.-\frac{7}{288}-\frac{49}{144}y^4\right)\alpha_6+\frac{y^{10}}{24}+\frac{25y^2}{384}+\frac{3y^6}{16}+\frac{25}{768}-\frac{5y^8}{16}+\frac{3y^4}{32}\right)\alpha_3+\left(\right.\\&\left.-\frac{7}{360}y^6-\frac{7}{16}y^2-\frac{7}{192}-\frac{7}{16}y^4+\frac{1}{240}y^8\right)\alpha_5\alpha_4+\left(-\frac{1}{40}y^8+\frac{3}{8}y^2+\frac{9}{32}y^4\right.\\&\left.+\frac{3}{64}\right)\alpha_5+\left(\frac{3}{56}y^2+\frac{11}{168}y^4+\frac{1}{252}y^6+\frac{1}{336}\right)\alpha_7\end{aligned} \quad \textbf{(13)}$$

$$diff5 := 0$$

```
> if number_polynomials >= 6 then
    Chung[6] := ((7/8748)*z^13-(1/524880)*z^17-(665/17496)*z^9-
  (805/972)*z^5-(1/65610)*z^15-(245/486)*z^7+(3115/1296)*z+
  (665/486)*z^3+(35/4374)*z^11)*alpha[3]^6+((-(49/2592)*z^11-
```

```
(4585/864)*z^3+(1/23328)*z^15+(595/7776)*z^9+(1225/864)*z^7+
(665/288)*z^5-(2485/288)*z-(7/7776)*z^13)*alpha[4]-(25/648)*z^9-
(65/72)*z^5+(5/16)*z^3+(35/24)*z-(1/3888)*z^15+(1/648)*z^13-
(35/48)*z^7+(47/1296)*z^11)*alpha[3]^4+((1/3240)*z^13+(2/405)*
z^11+(23/9)*z^3-(61/72)*z^5-(5/9)*z^7-(23/648)*z^9+(31/8)*z)*
alpha[5]*alpha[3]^3+(((475/96)*z^3-(1/5184)*z^13-(25/1728)*z^9-
(265/192)*z^5+(7/864)*z^11-(125/144)*z^7+(1445/192)*z)*alpha[4]
^2+((31/24)*z^7-(7/144)*z^11+(77/48)*z^5+(1/432)*z^13-(29/8)*z-
(1/36)*z^9-(53/48)*z^3)*alpha[4]+(-(10/9)*z^3-(1/810)*z^11+
(7/36)*z^5+(1/6)*z^7+(5/648)*z^9-(37/24)*z)*alpha[6]+(1/6)*z-
(1/36)*z^3-(35/144)*z^5-(1/144)*z^13-(1/18)*z^9-(5/12)*z^7+
(1/12)*z^11)*alpha[3]^2+((-4*z-(45/16)*z^3+(1/108)*z^9+(3/8)*z^7
-(1/720)*z^11+(7/12)*z^5)*alpha[5]*alpha[4]+((1/120)*z^11+(1/2)*
z^3+(1/120)*z^9+(3/2)*z-(17/40)*z^7-(21/40)*z^5)*alpha[5]+((1/3)
*z^3-(11/315)*z^7-(1/30)*z^5-(1/756)*z^9+(5/12)*z)*alpha[7])*
alpha[3]+((35/576)*z^5+(35/576)*z^7-(55/10368)*z^9+(1/10368)*
z^11-(425/384)*z-(835/1152)*z^3)*alpha[4]^3+((37/32)*z-(23/96)*
z^5-(1/576)*z^11+(7/144)*z^9-(11/48)*z^7+(103/192)*z^3)*alpha[4]
^2+(((7/12)*z^3+(5/6)*z-(7/180)*z^7-(1/60)*z^5+(1/540)*z^9)*
alpha[6]+(29/96)*z^7-(13/96)*z^9-(1/24)*z^3+(1/96)*z^11-(1/4)*z+
(5/32)*z^5)*alpha[4]+(-(13/600)*z^7-(1/800)*z^9+(51/160)*z-
(47/1200)*z^5+(29/120)*z^3)*alpha[5]^2+(-(1/90)*z^9+(1/15)*z^5-
(1/4)*z^3+(1/12)*z^7-(1/2)*z)*alpha[6]+(-(1/480)*z^5+(11/3360)*
z^7-(3/32)*z-(7/96)*z^3)*alpha[8]-(5/16)*z^7+(5/24)*z^9-(1/48)*
z^11;
  diff6 := expand(Chung[6]-p[6]);
end;
```